


ATARI

TIME * LOST

A COMPUTER ADVENTURE



que™

Timelost

Atari Version
(400, 800, & 1200XL Models)

Kris Austen Andrews
Arlan Keith Andrews, Sr.
Joseph C. Giarratano

Que Corporation
Indianapolis

Series concept: Joseph C. Giarratano and
Arlan Keith Andrews, Sr.

Story and Artwork: Kris Austen Andrews

Programs: Joseph C. Giarratano

Story consultant: Arlan Keith Andrews, Sr.

Lettering: Andy Andrews

Library of Congress Catalog Card Number: 83-61919

ISBN 0-88022-056-2

About the Authors

Arlan Keith Andrews, Sr., is a mechanical engineer at American Bell in Indianapolis. He received his Doctor of Science in Engineering degree from New Mexico State University and is a member of Science Fiction Writers of America. Kris Andrews, born in 1967 at Las Cruces, New Mexico, is currently a junior at Carmel High School, Carmel, Indiana. He is an honor student who desires to be a theoretical physicist. He is also a science fiction fan and likes role-playing games and computers.

Joseph C. Giarratano received his B.S. and M.S. degrees in physics from California State University at Los Angeles and his Ph.D. degree in physics from the University of Texas at Austin. He is the author of: *Foundations of Computer Technology*; *Modern Computer Concepts*; *BASIC: Fundamental Concepts*; *BASIC: Advanced Concepts*; the *Timex/Sinclair 1000 User's Guide*, Volumes 1 and 2; and the *Timex/Sinclair 1000 Dictionary and Reference Guide*.

Acknowledgment

We greatly appreciate the work of Tim Medlock in writing the Atari version of Battle at Stonehenge and Attack of the Slime Creatures; and Philip Oliver for writing Peril of the Pitdemons, Caverns of Carnage, Rockfall, and Escape.

Dedications

To my family:

Jane, Jenna, Melissa, and Anthony Giarratano

To the memory of
Erika Monique de Calonne

—*Joseph Giarratano*

To the family:

Joyce, Patty, Andy, Mandy, Sean,
Vivian Andrews Weed, Arlie Andrews,
and Mrs. Lizzie Van Ness

—*Kris and Arlan Andrews*

...and my Mom

—*Kris Andrews*

The characters in this book are completely fictional. Any similarity between these characters and any actual person, living or dead, is unintentional and purely coincidental.

Other *Timelost* Books by Que

Book:	ISBN No.	Date Available
<i>Timelost</i> (Timex/Sinclair 1000 Version)	0-88022-030-9	July, '83
<i>Timelost</i> (TI-99/4A Version)	0-88022-053-8	July, '83
<i>Timelost</i> (VIC-20 Version)	0-88022-054-6	July, '83

Other Timex/Sinclair Books by Que

Book:	ISBN No.	Date Available
<i>Timex/Sinclair 1000 User's Guide</i> , Vol. 1	0-88022-016-3	Currently
<i>Timex/Sinclair 1000 User's Guide</i> , Vol. 2	0-88022-029-5	Currently
<i>Timex/Sinclair 1000 Dictionary and Reference Guide</i>	0-88022-041-4	July '83

Note to the Reader

Timelost is designed so that it can be used in different ways. The cartoons can be read straight through as an episodic adventure story. Or, if you wish, you can choose the games that appeal to you the most, key the programs into your computer, and just play those games. Each game is self-contained. You don't have to read the cartoon section to understand how to play the game.

You may get the most enjoyment, however, by reading one episode and then playing the corresponding game at the back of the book. If you want to use the book this way, look for the "Play the game here" notation in the bottom right-hand corner of certain pages. At that point, simply refer to the Contents at the beginning of the book to find the page on which the appropriate game begins.

All game program listings have been thoroughly tested by the programmer(s) and work properly if typed correctly. If a problem occurs in playing any of the games, check carefully your entry of the listing.

Contents

Introduction

The Mystery of Stonehenge.....	1
The Slime Creatures.....	14
The Pitdemons	28
The Cavern Monsters	46
The Dungeon of the Lord-Wizard ..	59
Cave-In	72

Games

Battle at Stonehenge	85
Game Program Listing	85
Explanation of the Program	90
Some Changes You Might Like to Try	92
Attack of the Slime Creatures	93
Game Program Listing	94
Explanation of the Program	98
Some Changes You Might Like to Try	101
Peril of the Pitdemons.....	102
Game Program Listing	102
Explanation of the Program	109
Some Changes You Might Like to Try	113

Games *(cont'd)*

In the Caverns of Carnage	114
Game Program Listing	114
Explanation of the Program	123
Some Changes You Might Like to Try	125
Escape	126
Game Program Listing	127
Explanation of the Program	133
Some Changes You Might Like to Try	135
Rockfall!	136
Game Program Listing	136
Explanation of the Program	145
Some Changes You Might Like to Try	147

TIMELOST GAME CASSETTES!

For instant enjoyment of the Timelost game programs, order the Timelost cassette tapes for your computer. Each tape contains the programs exactly as they are found in the book. Play the computer games with all your favorite Timelost characters—without having to enter the programs yourself!

Please send me the following Timelost cassette(s):

Description	How Many?
Timelost cassette, VIC-20 format	_____
Timelost cassette, TI-99/4A format	_____
Timelost cassette, Atari format	_____

_____ cassettes x \$19.95 each = \$_____

Shipping & Handling (\$2.00 per item) = \$_____

TOTAL = \$_____

Method of Payment:

Check_____ MasterCard_____ VISA_____ Am. Exp._____

Credit Card Number: _____

Expiration Date: _____

Cardholder Name: _____

Ship To: _____

Address: _____

City: _____ State: _____ Zip: _____

Tear out and mail to:

Timelost Game Cassette
Que Corporation
7960 Castleway Drive
Indianapolis, IN 46250

Introduction

Welcome to the Universe of THE TIMELOST! In this new and exciting series, you'll follow the adventures of John Moore, a teenage computer whiz; his little sister, Erin; and their mysterious friend from the future. Join them as they battle against the evil Wizard, a Warlord of the Universe.

These books combine a story with programs that you can enter and run on an Atari 400, 800, 1200XL, and other compatible Atari computers. You follow the story and then get into the action with the game programs. All of the games were written especially to follow the adventures in the book.

These game programs are designed to be fun and educational. If you don't know how to program, you'll enjoy playing the games and may wish to learn about programming. If you do know how to program, or are learning how, the detailed explanations will show you how the games work and how you can change them and improve them.

The games in this book are designed to cover many different sound, color, and animation effects you can achieve with your computer. We know they'll be fun!

So, enter the fun world of computer games and the mysterious worlds of THE TIMELOST...


STANDING STRONG AND SILENT,
TOWERING OVER THE GRASSY PLAINS
OF SALISBURY, ENGLAND -- THE
MYSTERIOUS STONE CIRCLES OF
STONEHENGE!

IT IS HERE, IN THIS PLACE OF
ANCIENT WONDER, THAT THE JOURNEY
BEGINS... A JOURNEY FRAUGHT WITH
SCIENCE AND SAVAGES, MAGIC AND
MAYHEM, SPANNING THE FARTHEST
REACHES OF TIME AND SPACE.


FOR IT IS HERE THAT THE PRESENT
AND THE FUTURE SHALL MEET AND THE
UNSUSPECTING SHALL BECOME...

TIME! LOST!

LADIES AND GENTLEMEN OF THE
INTERNATIONAL COMPUTER CONFERENCE:
WELCOME TO STONEHENGE. A VISIT TO OUR
ANCIENT MONUMENT WILL ADD TO THE
ENJOYMENT OF YOUR TECHNICAL
DISCUSSIONS AT THE UNIVERSITY.



WE ARE STANDING AT THE EDGE OF THE LARGEST OF SEVERAL CIRCLES OF STONE... "MEGALITHS", WE CALL THEM. THE TALLEST STONES ARE TWENTY FEET HIGH, IN A CIRCLE 108 FEET IN DIAMETER. FURTHER OUT, SMALLER STONES LINE UP WITH THESE CIRCLES. SOME SCIENTISTS SAY THAT THE ALIGNMENTS POINT OUT THE BEGINNINGS OF THE SEASONS, AND THE RISING OF THE STARS.



WHATEVER THEY WERE USED FOR, THEY ARE FAR OLDER THAN THE PYRAMIDS OF EGYPT.

THE LATEST **MODERN** EXPLANATION, TAKING INTO ACCOUNT ALL OF THE CIRCLE ALIGNMENTS, IS THAT STONEHENGE IS ACTUALLY AN ASTRONOMICAL **COMPUTER**-- A COMPUTER THAT WAS USED TO PREDICT ECLIPSES AND PHASES OF THE MOON WITH ALMOST THE SAME PRECISION AS THE LATEST COMPUTERS OF OUR TIME.

IN THE TOUR GROUP, DRS. MARY AND PETER MOORE, COMPUTER SCIENTISTS, LISTEN INTENTLY, UNAWARE OF THE IMPENDING FATE OF THEIR CHILDREN...



THERE IS A LESS PLAUSIBLE, BUT MUCH MORE ROMANTIC, LEGEND WHICH CLAIMS THAT IN KING ARTHUR'S TIME, THE STONES WERE PUT IN PLACE BY A MAGICIAN...

... TEENAGER JOHN MOORE, HIGH SCHOOL COMPUTER WHIZ...



WOW!

... A SORCERER WE KNOW AS MERLIN.

...AND SEVEN-YEAR-OLD ERIN MOORE, DEVOTEE OF COMPUTER VIDEO GAMES.




MERLIN-? WAS HE LIKE JOHN...












IT WOULD SEEM TO BE A
DAY FOR SURPRISES!






A NIGHTMARE
ATTACKER...



A MYSTERIOUS
STRANGER...




CONFUSION RUNS
RAMPANT, BUT ONE
THING IS CLEAR---



--THEY ARE
BEING SAVED!

THE STRANGER FIRES HIS UNUSUAL WEAPON AGAIN AND AGAIN, BUT STILL THE DEMONS PRESS CLOSER!



ERIN AND JOHN CAN ONLY LOOK ON IN HORROR, KNOWING THAT THEIR VERY LIVES DEPEND UPON THE BATTLE'S OUTCOME!





Play the game here.

IT IS ANOTHER PLACE, ANOTHER TIME. THE DEMON IS GONE; THE MYSTERY OF STONEHENGE SEEMINGLY FAR BEHIND, REPLACED BY YET A DEEPER ENIGMA:

HOW, AND MORE IMPORTANTLY WHY, WERE THEY BROUGHT TO THIS RANK AND DISMAL PLACE IN A FORGOTTEN CORNER OF TIME?




THE SEETHING SWAMP SEEMS TO STRETCH FARTHER THAN THE YOUTHS' STUNNED MINDS CAN ACCEPT. AS FOR THE STRANGER, HIS THOUGHTS ARE HIS OWN.



THE PRESENT FACES THE FUTURE:
EACH A STRANGER IN THIS UNKNOWN
LAND, BROUGHT TOGETHER BY A
BIZARRE STROKE OF FATE --- OR
PERHAPS SOMETHING MORE

BUT WHILE THE
TIME BARRIER SEEMS TO
HAVE BEEN BROKEN, THE
TRAVELLERS ARE YET
SEPARATED BY A
DEEPER DIVISION:
THE LANGUAGE BARRIER.










ANYWAY, SUFFICE IT TO SAY THAT AT A CRUCIAL MOMENT I SUDDENLY FOUND MYSELF SWEEPED UP BY ONE OF THOSE TIME WARPS AND DEPOSITED AT STONEHENGE IN THE TWENTIETH CENTURY.... SAY, WAITAMINNIT!!



I WONDER IF -- MAYBE, JUST MAYBE
.... MAYBE CONOL -- NO, NO. I
DON'T THINK SO... IT
HAD TO BE A NATURAL WARP...
THERE'S NO WAY HE COULD HAVE
FOUND OUT...





... HEAVING FROM THE MUCK
ITS FIERCEST OF VENGEFUL
SPIRITS! THE SWAMP DOES
NOT LIKE TRESPASSERS...

... AND THE
WILL OF THE
SWAMP IS THE
LAW OF
THE LAND!



THE SLIME-DWELLERS
LURCH MENACINGLY CLOSER...



THESE THINGS JUST
KEEP COMING!



I'M HOLDING MY OWN,
BUT IT'S JUST A MATTER
OF TIME BEFORE --



AAACC HH!




THAT'S IT!
WARP OR NO
WARP, WE'VE
GOTTA GET
OUTTA HERE!!






Play the game here.








... BUT I THINK
I KNOW A WAY TA
FIND OUT.

"... SHOULD JUMBLE UP
OUR FRIEND'S THOUGHTS
ENOUGH TO DRIVE
HIM INTO THE OPEN!"



WHA---OH!




THIS QUASI-TELEPATHIC
THOUGHT TRANSLATOR,
SET ON HIGH....





WITH THE LITHE EASE OF AN ACROBAT,











YOU'RE IN THE KINGDOM OF
YRLAND. THAT'S THE HAMLET
OF MARSHSIDE DOWN
THERE.














WITHOUT WARNING
THE VERY GROUND
CAVES IN ON ITSELF,
OPENING A PIT
WITHOUT DEPTH...

... UNLEASHING A
SAVAGE SWARM
OF TUNNELLING
TERRORS BENT
ON ONE PURPOSE:


**GET
BACK!**






... THE
ABDUCTION
OF THE
UNLIKELY
QUARTET!











PLEASE, MISTER! YOU'VE GOT TO FIND MY BROTHER! YOU'VE GOT TO, MISTER!

CALL ME JACQUE. AND DON'T WORRY, I'VE GOT HIM ON MY SCANNERS....

...AND HE'S GONNA LEAD
ME TO WHOEVER'S BEHIND
ALL THIS...

MINUTES LATER...

...UH, DON'T YOU
THINK IT'S A
LITTLE DANGEROUS
TO COME BACK
IN HERE ?

...STUPID
LEECHES...

DON'T WORRY. MY POSITRON
BLASTER HASN'T FAILED ME YET...

...WISH
I HAD A
FORCE
FIELD
GENERATOR
...







WELL, THAT'S WHERE THE SIGNAL
STOPPED, SO THAT'S WHERE I'M
GOING!



ERIN, YOU STAY HERE
WITH D'REEN...

ER- I CAN'T WATCH HER! I CAN
BARELY LOOK AFTER MYSELF
IN THIS PLACE!



BESIDES, I WANNA STAY WITH YOU,
JACQUE! I WANNA HELP FIND JOHN!



IT'LL BE
DANGEROUS---
BUT OKAY.
BUT YOU'LL
HAFTA KEEP
QUIET, AND
YOU'LL HAFTA
DO WHAT I SAY, OR
ELSE I'LL LEAVE
YA THERE TA DIE
AND FIND MY OWN
WAY HOME!
GOT THAT?



Y- YESSIR...



...STAND BACK...



SCANNERS SHOW
THE HALL
IS CLEAN.



READY,
GIRL?

YEAH.

COMING, D'REEN?





NOT ME!

I MAY BE DUMB,
BUT I'M NOT
SUICIDAL!



... NOSSIR, NOT ME




HOPE I DIDN'T MAKE A MISTAKE...



... WISH MY SCANNERS WOULD STOP FRITZING...



JACQUE... FLUX'S PERKED UP...





... LIKE HE'S SEEN SOMETHING...

YEAH, I FEEL IT, TOO...



BLAST, I WISH MY SCANNERS
WORKED....



WITH A BESTIAL HOWL, THE VERY
WALLS SUDDENLY TRANSFORM
THEMSELVES INTO A CLAWING
MASS OF FANGS AND TALONS!

HERE ABIDE MONSTERS!

NO!!


HOLY—!



...AND THE FRAY IS ENTERED!
THE CAVERNOUS HALL IS FILLED
WITH THE INSANE HOWLS OF
THE ONCOMING GHOULS, THE
THUNDEROUS BOOMS OF
THE POSITRON BLASTER, AND
THE ECHOING SCREAMS OF
A HORRIFIED CHILD!!







Play the game here.


MEANWHILE, IN A
PLACE NOT FAR AWAY,
A BATTERED JOHN
STRUGGLES FROM A
RESTLESS SLUMBER...



TO FACE A
HARSHER
REALITY...





OOOOOH,...
..MY BACK...
WH-WHERE
..AM I...
?





... SOME KIND OF
JAIL ... DUNGEON...


... I-I WAS BEING
DRAGGED UNDER
--- BY THOSE
THINGS...























THE GIRL--AJA--LOOKING AT ME---



-- LIKE SHE KNOWS ME----

I'M--I'M STILL CONFUSED... WHAT
WOULD SOME POWERFUL WIZARD
WANT WITH ME ?


JOHN!

WHY I WANT YOU, MY BOY ?






... A FAR MORE THREATENING
DRAMA IS BEING PLAYED OUT
BENEATH THE SURFACE OF
THE SWAMP, NOT MANY
MILES AWAY!



FOR BLOOD'S SAKE, ERIN, STOP
SCREAMING! MY SCANNERS ARE
JAMMED ENOUGH AS IT IS!











-- J-JACQUE--?













THE CAVE WALLS BUCKLE UNDER THE
FORCE OF THE BLOW---











... WHILE THE LIVES OF TWO INNOCENT,
UNWILLING CHILDREN LAY HELPLESSLY
ON THE LINE!

... AND THOUGH
THEY MAY
BE
DEFIANT....

... AND THOUGH
THEY MAY
BE
SPIRITED....

.... THEY ARE STILL CHILDREN,
WHO HAVE UNEXPECTEDLY
BECOME ---

TIMELOST!

Battle at Stonehenge

How long can you defy the terrible Warbirds of the Wizard? Test your skill with a positron beam as you fight off the dreaded Warbirds in the Battle of Stonehenge. But beware: if a Warbird reaches the ground, the game is over.

GAME PROGRAM LISTING

```
100 ? CHR$(125):??:??:?"BATTLE AT
STONEHENGE"
110 ? :? :? :? :? "USE JOYSTICK PORT #1"
120 ? :? :? " PUSH FIRE BUTTON TO
    BEGIN PLAY"
130 IF STRIG(0)<>0 THEN GOTO 130
140 ? CHR$(125)
150 DIM PM1$(128),PM2$(128),PM3$(128),
    B$(3),PM0$(128)
160 SIZE1=13:SIZE2=10:SIZE3=9
170 DIM MAN$(SIZE1),BIRD$(SIZE2),
    MISL$(SIZE3),MAN2$(SIZE1)
180 DIM CLEAR$(128)
190 FOR ROWS=1 TO SIZE1
200 READ DOTS
210 MAN$(ROWS,ROWS)=CHR$(DOTS)
220 NEXT ROWS
```

```
230 DATA 125,85,125,25,255,188,152,  
    60,60,36,66,129,129  
240 FOR ROWS=1 TO SIZE1  
250 READ DOTS  
260 MAN2$(ROWS,ROWS)=CHR$(DOTS)  
270 NEXT ROWS  
280 DATA 125,85,125,25,255,188,152,  
    60,60,36,36,36,36  
290 FOR ROWS=1 TO SIZE2  
300 READ DOTS  
310 BIRD$(ROWS,ROWS)=CHR$(DOTS)  
320 NEXT ROWS  
330 DATA 0,0,0,153,90,60,24,0,0,0  
340 FOR ROWS=1 TO SIZE3  
350 READ DOTS  
360 MISL$(ROWS,ROWS)=CHR$(DOTS)  
370 NEXT ROWS  
380 DATA 0,1,1,1,1,1,0,0,0  
390 FOR ROWS=1 TO 128  
400 CLEAR$(ROWS,ROWS)=CHR$(0)  
410 NEXT ROWS  
420 A=4*(INT(PEEK(742)/4)-1)  
430 POKE 54279,A  
440 VSA=256*PEEK(135)+PEEK(134)  
450 BOA=256*PEEK(141)+PEEK(140)  
460 PM=256*A+512  
470 DISP=PM-BOA  
480 ADD=2  
490 FOR T=1 TO 4
```

```
500 PMHIGH=INT(DISP/256)
510 PMLOW=DISP-256*PMHIGH
520 POKE VSA+ADD,PMLOW
530 POKE VSA+ADD+1,PMHIGH
540 DISP=DISP+128:ADD=ADD+8
550 NEXT T
560 PM1$=CLEAR$:PM3$=CLEAR$
570 POKE 559,46:POKE 53277,3
580 POKE 53257,1:POKE 623,1
590 COLR=0:COLR2=12:COLR3=0
600 POKE 704,COLR1:POKE 705,COLR2:POKE
    706,COLR3:POKE 707,COLR1
610 GOSUB 1120
620 X1=125
630 PM2$=CLEAR$
640 X=RND(0):IF X<0.2 OR X>0.8 THEN 640
650 X2=X*240
660 POKE 53248,X1:POKE 53249,X2
670 Y1=90:Y2=25
680 PM1$(Y1)=MAN$:PM2$(Y2)=BIRD$
690 IF STRIG(0)=0 THEN GOTO 870
700 GOSUB 740
710 GOSUB 820
720 GOTO 690
740 C1=X1
750 IF STICK(0)<8 THEN X1=X1+3:GOTO 780
760 IF STICK(0)>8 AND STICK(0)<13
    AND X1>50 THEN X1=X1-3
770 IF C1=X1 THEN RETURN
```



```
780 IF X1/2=INT(X1/2) THEN PM1$=
    CLEAR$:POKE 53248,X1:PM1$(Y1)=
    MAN$:RETURN
790 PM1$=CLEAR$:POKE 53248,X1:
    PM1$(Y1)=MAN2$
800 RETURN
820 Y2=Y2+1.5
830 IF Y2>81 THEN 1070
840 POKE 53249,X2:PM2$(Y2)=BIRD$
850 RETURN
870 Y3=84
880 POKE 53250,X1:POKE 53278,0
890 PM3$(Y3)=MISL$:Y3=Y3-2
900 IF Y3<Y2-6 THEN GOTO 1040
920 IF PEEK(53261)<>0 THEN 950
930 GOTO 890
950 SCORE=SCORE+INT(Y2/10)
960 POSITION 1,1:? "SCORE:";SCORE
970 FOR J=-15 TO 15 STEP 3
980 SOUND 0,2,4,15-ABS(J):SOUND 1,
    10,10,15-ABS(J)
990 FOR K=1 TO 5:NEXT K
1000 NEXT J
1010 PM3$=CLEAR$
1020 GOTO 630
1040 PM3$=CLEAR$
1050 GOTO 690
1070 POSITION 6,4:? " Wish to play again (Y/N)";
1080 TRAP 1080:INPUT B$
```

```
1090 IF B$="Y" THEN RUN
1100 POKE 53277,0: ? CHR$(125);:END
1120 POSITION 9,10: ? "Δ";CHR$(18);CHR$(23);
    CHR$(18);CHR$(18);CHR$(23);CHR$(18);
    "ΔΔΔΔΔΔΔΔΔΔΔΔ";
1125 ? CHR$(18);CHR$(23);CHR$(18);CHR$(18);
    CHR$(23);CHR$(18);"Δ"
1128 ? "ΔΔΔΔΔΔΔΔΔΔ";CHR$(124);"ΔΔ";
    CHR$(124);"ΔΔΔΔΔΔΔΔΔΔΔΔΔΔΔΔ";
    CHR$(124);"ΔΔ";CHR$(124)
1130 ? "ΔΔΔΔΔΔΔΔΔΔ";CHR$(124);"ΔΔ";
    CHR$(124);"ΔΔΔΔΔΔΔΔΔΔΔΔΔΔΔΔΔΔ";
    CHR$(124);"ΔΔ";CHR$(124)
1150 RETURN
```

Variables Used in this Game Include:

PM1\$=1st player
PM2\$=2nd player
PM3\$=3rd player
PM4\$=4th player
MAN\$=Name for 1st player
MAN2\$=Name for 2nd player
BIRD\$=Name for bird
MISL\$=Name for missile
B\$=Answer to "play again ?" question
CLEAR\$ =Blank string overlay
X1=Horizontal position players 1 & 4
X2=Horizontal position player 2
Y1=Vertical position players 1 & 4
Y2=Vertical position player 2
Y3=Vertical position player 3, the missile
Strig(0)=Fire button joystick 1
Δ=Space

Explanation of the Program

- 100 Directions 100-120
- 130 If fire button pushed then continue
- 140 Clear screen
- 150 Reserves 128 spaces for players height
- 160 Sets players vertical size
- 200 Reads in shape of man 1 into MAN\$ string
200-220
- 230 Shape of man 1
- 240 Reads in shape of man into MAN2\$ string
240-270
- 280 Shape of man 2
- 290 Reads in shape of bird into BIRD\$ string
290-320
- 330 Shape of bird
- 340 Reads in shape of missile into MISL\$ string
340-370
- 380 Shape of missile
- 390 Reads in CLEAR\$ string which is a blank
string 390-410
- 420 Tells the computer to use player/missile
graphics 420-550
- 490 Set upper limit to the number of players you
wish to use, 4 is max.
- 560 Clears out any stray data
- 570 Sets P/M graphics to double line
resolution, turns on P/M graphics
- 580 Sets size register of bird to double width

590 Sets colors of the players, 0=black,
12=white, 244=dark blue
600 Sets color values into player registers
610 Gosub to draw stonehenge background
620 Horizontal position of player 1
630 Clears player 2
640 Picks a random x value between .2 and .8
650 $X2 = X$ times 240
660 Puts horizontal positions of players 1 & 2
into registers
670 Vertical positions of players 1 & 2
680 Draws vertical positions of players 1 & 2 on
the screen
690 If fire button is pushed goto fire routine
700 Goto move man subroutine
710 Goto move bird subroutine
720 Repeat sequence
750 If joystick "right" then increase X1 value
760 If joystick "left" then decrease X1 value
820 Y2 value is increased causing bird to drop
down
830 Tests position of bird to see if it is too low
840 Display bird at X2 and Y2 location
870 Vertical position of missile
880 Displays missile at X1 location and sets
collision registers to 0
890 Displays missile at Y3 location and then
decreases Y3 by 2
900 If missile is 6 spaces beyond bird then clear
missile

920 Checks player 1 to player collision register
to see if a hit
950 Score = score + vertical position of bird
970 Loop for sound of hit, which increases by 3
each time
1010 Clears missile from screen
1070 Prints question on screen at position X=6,
Y=4
1080 Waits for answer, if any error it goes back
and waits again
1090 If answer if Y then play game again
1100 If answer is N then clear players from
screen and end
1120 Starts drawing background at X=9, Y=10
Otherwise the game ends (1350).

Some Changes You Might Like to Try:

1. Make the Warbird move in a zigzag pattern down the screen, instead of straight down.
2. Penalize the player for missed shots. For example, give 10 points for every Warbird the man hits and take away 5 points for every one missed.
3. Let the Warbirds appear at random locations down the screen as well as horizontally.
4. Allow multiple Warbirds on the screen, instead of just one at a time.

Attack of the Slime Creatures

Hold off the Slime Creatures! How long can you avoid the terrors as they come closer and closer?

In this game, the Slime Creatures appear on the right and move toward Jacque on the left. The Slime Creatures appear and disappear as they swim underwater toward Jacque. Then suddenly they appear for a moment as they surface to attack Jacque.

You'll hear a tone proportional to the time they will appear on the screen. A longer tone means the Slime Creature will be visible a longer time. If you shoot and miss, the Creature can move up to twice as far on its next move toward you. The pitch of the tone increases as the Creature gets closer to you.

GAME PROGRAM LISTING

```
100 ? "↵":? :? :? :? " ATTACK OF THE
    SLIME CREATURES"
110 ? :? :? :? :? :? " PRESS THE SPACE BAR
    TO FIRE BLASTER"
120 ? :? :? "PRESS ANY KEY TO BEGIN"
130 POKE 764,255
140 IF PEEK(764)=255 THEN 140
150 ? "↵"
160 DIM PM1$(128),PM2$(128),PM3$(128),M$(1)
170 SIZE1=13:SIZE2=9:SIZE3=1
180 DIM MAN$(SIZE1),SLIME$(SIZE2),
    MISL$(SIZE3)
190 DIM CLEAR$(128)
200 FOR ROWS=1 TO SIZE1
210 READ DOTS
220 MAN$(ROWS,ROWS)=CHR$(DOTS)
230 NEXT ROWS
240 DATA 120,72,120,51,254,120,48,120,
    120,132,132,132,132
250 FOR ROWS=1 TO SIZE2
260 READ DOTS
270 SLIME$(ROWS,ROWS)=CHR$(DOTS)
280 NEXT ROWS
290 DATA 62,127,207,255,207,15,239,126,60
300 READ DOTS
310 MISL$(1)=CHR$(DOTS)
320 DATA 248
```

```
330 FOR ROWS=1 TO 128
340 CLEAR$(ROWS,ROWS)=CHR$(0)
350 NEXT ROWS
360 A=4*(INT(PEEK(742)/4)-1)
370 POKE 54279,A
380 VSA=256*PEEK(135)+PEEK(134)
390 BOA=256*PEEK(141)+PEEK(140)
400 PM=256*A+512
410 DISP=PM-BOA
420 ADD=2
430 FOR T=1 TO 3
440 PMHIGH=INT(DISP/256)
450 PMLOW=DISP-256*PMHIGH
460 POKE VSA+ADD,PMLOW
470 POKE VSA+ADD+1,PMHIGH
480 DISP=DISP+128:ADD=ADD+8
490 NEXT T
500 PM1$=CLEAR$:PM2$=CLEAR$:
    PM3$=CLEAR$
510 POKE 559,46:POKE 53277,3
520 COLR1=0:COLR2=12:COLR3=72
530 POKE 704,COLR1:POKE 705,COLR2:
    POKE 706,COLR3
540 GOSUB 850
550 POKE 53248,60
560 PM1$(60)=MAN$
570 POSITION 1,1:? " SCORE: ";SCORE
580 DT=1
```



```
590 X2=190-INT(RND(0)*25)
600 FOR JO=1 TO X2:NEXT JO
610 TT=5+INT(RND(0)*X2/3)
620 FOR BEEPS=1 TO INT(TT)
630 SOUND 0,X2-65,10,22-X2/10
640 NEXT BEEPS
650 SOUND 0,0,0,0
660 POKE 53249,X2:PM2$(60)=SLIME$
670 POKE 764,255
680 IF PEEK(764)<>255 THEN 710
690 TT=TT-1:IF TT<=0 THEN 800
700 GOTO 680
710 POKE 53250,65:PM3$(63)=MISL$
720 FOR BLAST=70 TO X2+10 STEP 3
730 POKE 53278,0
740 TT=TT-1
750 IF TT<=0 THEN PM2$=CLEAR$
760 PM3$=CLEAR$:POKE 53250,BLAST:
    PM3$(63)=MISL$
770 IF PEEK(53261)<>0 AND TT>0 THEN
SCORE=SCORE+INT(200-X2):
    GOSUB 890:GOTO 570
780 NEXT BLAST
790 DT=2
800 PM3$=CLEAR$
810 X2=X2-INT(ABS(20*RND(0)+8))*DT
820 DT=1
830 IF X2<=70 THEN X2=70:GOTO 940
```

```
840 PM2$=CLEAR$:GOTO 600
850 FOR Z=4 TO 22
860 POSITION 0,Z:? " "
870 NEXT Z
880 RETURN
890 PM2$=CLEAR$:PM3$=CLEAR$
900 SOUND 0,36,12,15
910 FOR J=1 TO 100:NEXT J
920 SOUND 0,0,0,0
930 RETURN
940 POKE 53249,70:PM2$(60)=SLIME$
950 POSITION 10,18;? " GOOD SHOOTING "
960 POSITION 2,2:? "ANOTHER GAME "
970 POSITION 2,3:? "(Y=YES N=NO) ";
980 TRAP 980:INPUT M$
990 IF M$="N" THEN 1010
1000 ? "↵":CLR:GOTO 160
1010 PM1$=CLEAR$:PM2$=CLEAR$:? "↵"
1020 END
```

Variables Used in this Game Include:

PM1\$=1st player
PM2\$=2nd player
PM3\$=3rd player
MAN\$=Name of man
SLIME\$=Name of slime creature
MISL\$=Name of missile
M\$=Answer to "play again ?"
CLEAR\$=Blank string overlay

TT=Time creature appears

X2=Horizontal position of 2nd player, slime creature

COLR=Color for players

DT=Single/double multiplier for slime creature position

↵=Escape, Ctrl, Clear

Explanation of the Program

100 Directions 100-120

130 Clears keyboard character register

140 Reads any key pushed

150 Clears screen

160 Sets aside 128 spaces high for each player

170 Sets size of each player height

200 Read shape of man into MAN\$ string
200-230

240 Shape of MAN\$

250 Read shape of slime creature into SLIME\$
string 250-280

290 Shape of SLIME\$

300 Read shape of missile into MISL\$ 300-310

320 Shape of MISL\$

330 Reads in CLEAR\$ string which is a blank
string 330-350

360 Tells computer to use player/missile
graphics 360-490

430 Set upper variable to number of players
being used, 4 is max. 500 Clears out
everything

- 510 Sets P/M graphics to double line resolution
and turns on P/M graphics
- 520 Sets colors of players, 0=black, 12=white,
72=yellow
- 530 Puts colors into player registers
- 540 Draws boundary line on playing screen
- 550 Horizontal position of 1st player into player
register
- 560 Vertical position of 1st player as it is drawn
on screen
- 570 Print scoreboard on screen at position X=1,
Y=1
- 580 Sets multiplier to 1
- 590 Picks random X2 for starting position of
SLIME\$
- 600 Delay counter
- 610 Picks length of time slime creature appears
- 620 Starts tone which is proportional to time
slime creature appears
- 630 Sound
- 650 Turns sound off
- 660 Puts X2 value into 2nd player register and
displays it on screen at Y=60
- 670 Clears keyboard character register
- 680 Checks to see if a key is pushed to fire
blaster
- 690 Decreases time by 1
- 710 Puts 65 into horizontal register of 3rd
player and displays at Y=63

720 Starts loop to move blaster missile to the right
730 Clears the collision registers
750 After time is decreased if time is equal to zero creature disappears
760 Clear missile position and move missile to next location
770 Checks player collision register to see if missile hits creature
790 Multiplier is double since shot missed creature
800 Clear missile
810 Determine new location of creature
830 If horizontal position X2 is less than or equal to 70 then too close
850 Start loop to draw boundary line
860 Use inverse space character
890 Clear players off screen
900 Sound for a hit
910 Delay loop
920 Turns sound off
940 Put creature at boundary line of the play screen
950 Prints "Good Shooting"
980 Waits for answer to question
1000 If answer is Y then clear screen and start over
1010 If answer is N then clear all players and end

Some Changes You Might Like to Try:

1. Allowing multiple Slime Creatures to appear and disappear. You would allow Jacque to move on the screen to position him under the creature you want to shoot.
2. Limiting the number of Positron Blasts and Slime Creatures per game. For example, allow 20 Slime Creatures and 20 Positron Blasts to see if Jacque can hit them all. You should also display the Positron Blasts remaining.

Peril of the Pitdemons

Now, you must fight off a fierce attack by underground creatures of the Wizard! This time, the Pitdemons are after John, Erin, Jacque, and their new friend D'reen. In this game, you control Jacque's blaster as he fights against the swarming demons from the dark pit.

If any demon reaches the top line on which Jacque stands, you lose. Be especially careful when the demons reach the rocks bordering the pit. The demons are invisible in the rocks, so don't forget the demons are there.

GAME PROGRAM LISTING

```
1 REM *** PERIL OF THE PITDEMONS ***  
2 GRAPHICS 0:POKE 752,1:POSITION 9,4:PRINT  
  "INITIALIZING"  
3 REM CLEAR KEYBOARD,SET MARGINS  
4 POKE 764,255:POKE 82,1:POKE 83,39
```

```
5 SW=30:W=SW-2:BR=23
6 JJR=4:MAXM=7:RAD
7 DIM JJ$(2),M$(1),L$(SW),BL$(SW)
10 DIM HIGHNM$(10),M(SW)
20 STDSET=57344:CHBASE=PEEK(15)+32
30 CHBASE=4*INT(CHBASE/4)+4
40 CHADR=CHBASE*256
50 FOR X=0 TO 767
60 POKE CHADR+X,PEEK(STDSET+X)
70 NEXT X
80 P=CHADR+24
90 READ X:IF X>-1 THEN POKE P,X:
    P=P+1:GOTO 90
95 GOSUB 8000
150 TC=0:SCORE=0:M=37:M$=CHR$(M)
151 JJ$="$#"
155 FOR X=1 TO SW
157 L$(X,X)=CHR$(18)
160 NEXT X
170 OPEN #1,4,0,"S:":GOSUB 1000
180 POKE 752,1
185 JJP=W/2:LJJP=JJP:MPR=JJR
190 FOR X=1 TO SW
192 BL$(X,X)=" "
195 NEXT X
200 FOR X=1 TO W
205 M(X)=0:NEXT X
250 POSITION 1,JJR+1:PRINT L$;
```



```
255 POKE 752,1
257 Y=0
260 FOR X=JJR+2 TO 12
262 YL=Y
265 Y=INT(SW/2*SIN((X-JJR-2)*1.5708/(10-JJR)))
267 COLOR (8)
270 PLOT SW/2-YL,X:DRAWTO SW/2-Y,X
272 COLOR (10)
275 PLOT SW/2+YL,X:DRAWTO SW/2+Y,X
280 NEXT X
285 FOR X=13 TO BR
287 COLOR (8):PLOT 1,X
290 COLOR (10):PLOT SW,X
293 NEXT X
295 POSITION JJP,JJR:? JJ$;
300 X=INT(W*RND(0))+2:M(X)=BR:NM=1
305 POSITION X,BR:? M$;
400 LCNT=LCNT+1
410 GOSUB 800
420 IF RND(0)>0.5 OR NM=0 THEN 500
422 Y=1
425 FOR X=1 TO INT(NM*RND(0))
427 Y=Y+1
430 IF M(Y)=0 THEN 427
435 NEXT X
440 POSITION Y,M(Y):GET #1,X
442 X=X*(X<>ASC(M$))+32*(X=ASC(M$))
445 POSITION Y,M(Y):PUT #1,X
```

```
450 M(Y)=M(Y)-1-INT(1.01*RND(0))
451 SOUND 0,64*(ML-M(Y)),10,(M(Y)-23)/
    (JJR-24)*14+1:FOR T=0 TO 6:NEXT T:
    SOUND 0,0,0,0
452 IF M(Y)<JJR THEN M(Y)=JJR
453 IF MPR>=M(Y) AND MPC=Y AND
    MPR<M(Y)+3 THEN GOSUB 1775:
    GOTO 500
455 POSITION Y,M(Y):GET #1,X
457 POSITION Y,M(Y):PUT #1,X
458 IF X=32 THEN POSITION Y,M(Y):? M$;
460 IF M(Y)=JJR THEN 900
500 IF RND(0)>0.5 OR NM=MAXM THEN 555
510 Y=1
520 FOR X=1 TO INT((W-NM)*RND(0))
525 Y=Y+1
530 IF M(Y)<>0 THEN 525
540 NEXT X
542 SOUND 0,64+Y*3,10,15
545 M(Y)=BR:POSITION Y,BR:? M$;
547 SOUND 0,0,0,0
550 NM=NM+1
555 POKE 752,1
580 IF MPR=JJR THEN 600
583 POSITION MPC,MPR:PRINT " ";
585 MPR=JJR*(MPR=BR)+(MPR+1)*(MPR<BR)
587 GOSUB 1700
600 POSITION 2,0:? "SCORE: ";SCORE;
```

```
690 GOTO 400
800 A=PEEK(764)
802 IF A=255 THEN RETURN
803 POKE 764,255
807 LJJP=JJP
810 JJP=JJP-(A=54)+(A=55)
812 IF JJP<1 THEN JJP=1
813 IF JJP>SW-2 THEN JJP=SW-2
815 IF JJP<>LJJP THEN POSITION
    LJJP,JJR:PRINT " ";POSITION JJP,JJR:
    ? JJ$;
820 IF A<>33 OR MPR<>JJR THEN RETURN
825 MPR=JJR+2:MPC=JJP+1
839 GOTO 1710
900 FOR X=0 TO 2:POSITION 1,X:? BL$;
901 NEXT X
903 POSITION 10,1:? "***** M U N C H *****";
905 POKE 752,0:POKE 764,255
907 IF HIGHNM$="" THEN 910
908 POSITION 1,0:? "HIGH SCORE: ";HIGHSC;"
    BY ";HIGHNM$;
910 POSITION 12,2:? "*** GAME OVER ***"
911 FOR T=1 TO 300:NEXT T
912 FOR X=3 TO 6:POSITION 1,X:? BL$;
915 NEXT X:CLOSE #1
920 IF SCORE<HIGHSC THEN 960
925 HIGHSC=SCORE
930 POSITION 14,3:? "GOOD SHOOTING!"
```

```
940 POSITION 2,5:? "NEW HIGH SCORE.  
    YOUR NAME";:INPUT HIGHNM$  
950 POSITION 2,5:? BL$;  
960 POSITION 2,1:? BL$;  
962 POSITION 2,1:? " HIGH: ";HIGHSC;  
965 IF HIGHNM$<>" " THEN ? " (BY ";HIGHNM$;");  
970 POSITION 2,5:? "CARE FOR ANOTHER  
    GAME";:INPUT M$:IF M$="Y" THEN  
    GOSUB 1000:GOTO 150  
  
980 GRAPHICS 0:END  
999 END  
1000 GRAPHICS 0  
1010 POKE 756,CHBASE  
1015 SETCOLOR 2,1,2:SETCOLOR 1,0,15  
1020 RETURN  
1135 M(MPC)=0:NМ=NМ-1  
1700 IF MPR=JJR THEN RETURN  
1710 POSITION MPC,MPR:GET #1,X  
1720 POSITION MPC,MPR:PUT #1,X  
1725 IF X=8 OR X=10 THEN MPR=MPR+1:  
    GOTO 1710  
1730 IF X<>32 THEN 1770  
1735 POSITION MPC,MPR:PRINT ". ";:RETURN  
1770 IF X<>M THEN MPR=JJR:RETURN  
1775 SCORE=SCORE+BR-MPR  
1780 M(MPC)=0:NМ=NМ-1  
1790 POSITION MPC,MPR:? " *";  
1800 SOUND 0,4,8,15:FOR T=0 TO 3:NEXT T
```



```
1810 SOUND 0,8,8,15:FOR T=0 TO 3:NEXT T
1820 SOUND 0,0,0,0
1830 POSITION MPC,MPR:? " ";
1840 MPR=JJR:RETURN
2090 DATA 24,60,24,126,90,90,36,36
2130 DATA 24,60,24,60,90,90,36,36
2500 DATA -1
8000 GRAPHICS 0:POKE 82,4
8005 POSITION 6,1
8010 ? "*** PERIL OF THE PITDEMONS ***"
8100 ?
8110 ? "PRESS < TO MOVE LEFT OR"
8120 ? "> TO MOVE JACQUE AND ERIN RIGHT"
8130 ?
8135 ? "PRESS SPACE-BAR TO FIRE POSITRON"
8137 ? "BLASTER SHOTS TO HIT MONSTERS"
8138 ?
8140 ? "THE CLOSER THE MONSTER TO"
8150 ? "JACQUE WHEN FINALLY HIT, THE"
8160 ? "GREATER NUMBER OF POINTS ADDED"
8930 ?
8940 ? "PRESS ANY KEY TO BEGIN"
8945 A=PEEK(764):IF A=255 THEN 8945
8950 POKE 764,255:POKE 82,1
9000 GOTO 1000
```

Variables Used in this Game Include:

BL\$=String of blanks

BR=Bottom screen row

CHADR=Absolute memory address of alternate characters
CHBASE=Memory page of alternate characters (*256 = CHADR)
HIGHSC=Current high score
HIGHSC\$=Current high scorer's name
JJP=Jacque/Erin position (column)
JJR=Couple row
JJ\$=Characters for Erin/Jacque
L\$=Cavern top string
LCNT=Total # of game cycles
MAXM=Maximum # of monsters displayed at any one time
M(c)=Monster array. "c" = Screen column, M(c)=Screen row, (M(c)=0 indicates no monster in column "c")
M=ATASCII code of monster
M\$=Monster character
MPC=Missile column
MPR=Missile row (=JJR if no missile)
NM=Current number of monster\$
P=Miscellaneous variable
SCORE=Obvious; the score
STDSET=Absolute memory address of standard character set
SW=Screen width
W=Cavern width
X, Y=Miscellaneous variables

Explanation of the Program

- 2 Set graphics mode, turn off cursor
- 4 Clear keyboard, set margins

5-6 Set variables; "RAD" sets radian trig mode

7-8 Dimension strings and arrays

20-90 Set up alternate character set

95 Display instructions

150-151 Set variables (JJ\$ = Erin/Jacque
characters)

(M=Character code of monster)

155-160 Set up inverse blank string

170 Open "file" to access screen

180 Wipe out cursor

185 Set initial couple position

190-195 Set up string of spaces

200-205 Initialize monster array (=0 means no
monster)

250 Print cavern top

255 Re-erase cursor

257-280 Draw top arch of pit

260 "X" represents screen row

262 Save current "Y" value

265 Set "Y" equal to function that will determine
a circular arch. Determines distance of
"arch" from center of screen at screen row
"X"

267 Set up left cavern character

270 Plot left line from previous "Y" to current
"Y"

272 Set up right cavern character

275 Plot right line from previous "Y" to current
"Y"

280 Loop until arch drawn
285-293 Draw vertical portion of pit
295 Print Jacque/Erin
300 Set up first monster
305 Display first monster
400 Increment cycle counter
410 Perform key action
420 50/50 chance of monster moving in cycle
422-435 Search for monster to move
440 Get char. from screen at monster position
 ("GET#1,X" will blank screen character)
442 Set "X" equal to 32 if monster at screen
 pos.
445 Redisplay; blanked if monster was visible
450 Move monster up screen 1 to 2 rows
 (random select)
451 Sound effect, volume proportional to
 distance up screen
452 Set monster row to JJR if less than JJR
 (Jacque/Erin row)
453 If monster "hopped" over bullet, shoot it
455 Get char. from screen at new monster
 position
457 (Put char. back onto screen because of
 destructive "GET" command)
458 If blank, print monster at new position
460 If monster in Jacque/Erin row, all is lost
500-555 50/50 chance of new monster
 generated if number existent less than
 maximum

- 510-540 Search for randomly selected empty column
- 542 Start sound effect; pitch proportional to column
- 545 Print newly created monster
- 547 End sound effect
- 580-587 Update missile position if required
- 583 Clear missile from screen
- 585 Clear missile (by setting MPR=JJR) if at bottom row, else update it by one
- 587 Call to check to missile hit
- 600 Print score
- 690 ...and loop once more
- 800-802 Peek byte from keyboard, return if 255 (no keys)
- 803 Re-initialize keyboard for next stroke
- 807-815 Adjust player's position if necessary
- 820-839 Fire missile if necessary and possible
- 900-999 End of game, print high score, ask for name if new high score, repeat game if requested
- 1000-1020 Initialize screen
- 1010 Set up alternate character map base
- 1015 Set up appropriate colors
- 1710-1735 Check for missile hit
- 1770-1840 Missile hit, do update
- 1770 If missile hit non-monster, no score update

1775 Update score (max # of points if monster directly below Jacques/Erin, no points if at bottom)
1780 Clear monster at that column, decrement monster counter
1790 Print explosion character (inverted)
1800-1820 Sound effects of explosion
1830 Erase explosion character
1840 Clear missile and return
2090 Data for "Jacques" character
2130 "Erin"
2170 "Monster"
2500 Data statement terminator
8000-9000 Instructions

Some Changes You Might Like to Try

1. Let a Warbird attack from above while the Pitdemons are climbing up from below.
2. Add some different types of creatures climbing up and give different points for them.
3. Let some of the Pitdemons go invisible and give extra points for hitting them.

In the Caverns of Carnage

Jacque and Erin are in the fearsome and eerie Caverns of Carnage, attacked by the savage Crusher beasts. How long can Jacque and Erin survive the Chase that no one has ever gotten through before? How far can you get them through the Caverns?

You can use Jacque's Positron Blaster to destroy the loathsome Crushers and score points for each one disintegrated. You get more points the closer you are to a Crusher when you blast it. But watch out—if a Crusher gets Jacque or Erin from the front or they crash into a wall or Crusher, the game is over.

GAME PROGRAM LISTING

```
1 REM *** CAVERNS OF CARNAGE ***
2 GRAPHICS 0:POKE 752,1:POSITION 9,4:
  PRINT "INITIALIZING"
3 REM CLEAR KEYBOARD, SET MARGINS
4 POKE 764,255:POKE 82,1:POKE 83,39
5 SW=37:W=5:ML=SW-W-2
```

```
6 JJR=5:WM1=W-1
7 DIM JJ$(2),M$(1),L$(SW),MS$(W)
8 DIM HIGHNM$(10),BL$(SW),MSS$(W)
10 REM
12 REM SET UP SPECIAL CHARACTERS
14 REM
20 STDSET=57344:CHBASE=PEEK(15)+28
30 CHBASE=4*INT(CHBASE/4)+4
40 CHADR=CHBASE*256
50 FOR X=0 TO 767
60 POKE CHADR+X,PEEK(STDSET+X)
70 NEXT X
80 P=CHADR+24
90 READ X:IF X>-1 THEN POKE P,X:P=P+1:
    GOTO 90
95 GOSUB 8000
100 REM -----
110 REM TC = TOTAL # GAME CYCLES
112 REM JJP = COUPLE'S POSITION
114 REM LJJP = COUPLE'S LAST POS.
131 REM MPR = MISSILE ROW
132 REM MPC = MISSILE COL.
135 REM LMPR = LAST MISSILE ROW
136 REM LMPC = LAST MISSILE COL.
149 REM -----
150 TC=0:SCORE=0:M=37:M$=CHR$(M)
151 JJ$="$#"
152 REM -----
```

```
155 FOR X=1 TO SW
157 L$(X,X)=CHR$(32+128)
160 NEXT X
161 REM -----
165 L=ML*RND(0)
167 REM -----
170 OPEN #1,4,0,"S:":GOSUB 1000
171 REM -----
180 POKE 752,1
182 REM -----
185 JJP=L+W/2:LJJP=JJP:MPR=JJR
187 REM -----
190 FOR X=1 TO SW
192 BL$(X,X)=" "
195 NEXT X
197 REM -----
200 FOR X=1 TO W
205 MSS$(X,X)=" "
207 NEXT X
209 REM -----
300 LCNT=LCNT+1
497 REM
498 REM GENERATE LINE
500 REM
502 IF ABS(LCNT/3-INT(LCNT/3))>1.0E-05
    THEN GOSUB 800:GOTO 580
510 L=L+SGN(50-100*RND(0))
520 L=L*(L>0)*(L<ML)+(L<1)+ML*(L>=ML)
```



```
525 MS$=MSS$
530 FOR X=1 TO 4
532 POSITION 2,X:? BL$;
535 NEXT X
545 I=1+WM1*RND(0):MS$(I,I)=M$
547 GOSUB 1500:POSITION 2,23
550 ? L$(1,L);MS$;L$(1,SW-L-W)
555 POKE 752,1
557 REM
558 REM CHECK FOR JJ COLLISION
559 REM
560 GOSUB 1800
572 REM
573 REM CHECK FOR MISSILE COLLISION
574 REM
575 IF MPR<>JJR THEN GOSUB 1710
577 REM
578 REM UPDATE MISSILE POS.
579 REM
580 IF MPR=JJR THEN 600
583 POSITION MPC,MPR:PRINT " ";
585 MPR=JJR*(MPR=23)+(MPR+1)*(MPR<23)
587 GOSUB 1710
590 REM
600 POSITION 2,1:? "SCORE: ";SCORE;
615 IF HIGHNM$<>"" THEN ? "
    (BY ";HIGHNM$;")";
690 GOTO 300
```

```
698 REM
699 REM -----
800 A=PEEK(764)
802 IF A=255 THEN RETURN
803 POKE 764,255
805 REM ADJUST PLAYER'S POSITION
807 LJJP=JJP
810 JJP=JJP-(A=54)+(A=55)
812 IF JJP<3 THEN JJP=3
813 IF JJP>SW-2 THEN JJP=SW-2
815 IF JJP<>LJJP THEN POSITION LJJP,JJR:
    PRINT " ";GOTO 1800
817 REM
818 REM FIRE MISSILE?
819 REM
820 IF A<>33 OR MPR<>JJR THEN RETURN
825 MPR=JJR+1:MPC=JJP+1
839 GOTO 1710
840 REM -----
897 REM
898 REM JJ HIT SOMETHING - DOOM
899 REM
900 FOR X=1 TO 6:POSITION 2,X
901 CLOSE #1
902 PRINT BL$;NEXT X
905 POKE 752,0:POKE 764,255
910 POSITION 12,1:? "*** GAME OVER ***"
920 IF SCORE<HIGHSC THEN 960
```

```
925 HIGHSC=SCORE
930 POSITION 14,3:? "GOOD SHOOTING!"
940 POSITION 2,5:? "NEW HIGH SCORE.
    YOUR NAME";INPUT HIGHNM$
950 POSITION 2,5:? BL$;
960 POSITION 2,1:? BL$;
962 POSITION 2,1:? " HIGH: ";HIGHSC;
965 IF HIGHNM$<>" " THEN ? "
    (BY ";HIGHNM$;"");
970 POSITION 2,5:? "CARE FOR ANOTHER
    GAME";INPUT M$:IF M$="Y"
    THEN GOSUB 1000:GOTO 150
980 GRAPHICS 0:END
999 END
1000 GRAPHICS 0
1010 POKE 756,CHBASE
1015 SETCOLOR 2,1,5:SETCOLOR 1,0,15
1020 RETURN
1097 REM
1098 REM MISSILE HIT?
1099 REM
1100 POSITION MPC,MPR:GET #1,X
1110 POSITION MPC,MPR:PUT #1,X
1115 IF X=32 THEN RETURN
1120 IF X<>M THEN MPR=JPR:RETURN
1125 SCORE=SCORE+16-(MPR-JJR)
1130 POSITION MPC,MPR:PUT #1,32
1140 MPR=JJR
1190 RETURN
```

```
1199 REM
1500 POSITION JJP,JJR:? " ";
1510 IF MPR>JJR THEN POSITION MPC,MPR:? " ";
1520 RETURN
1600 POSITION JJP,JJR:PRINT JJ$;
1610 IF MPR>JJR THEN POSITION
      MPC,MPR:PRINT ". ";
1690 RETURN
1697 REM
1698 REM CHECK FOR MISSILE COLLISION
1699 REM
1710 POSITION MPC,MPR:GET #1,X
1720 POSITION MPC,MPR:PUT #1,X
1730 IF X<>32 THEN 1770
1735 POSITION MPC,MPR:PRINT ". ";RETURN
1740 REM
1750 REM MISSILE HIT SOMETHING
1760 REM
1770 IF X=M THEN GOSUB 1125:GOSUB 6000
1775 MPR=JJR
1780 RETURN
1797 REM
1798 REM CHECK FOR JJ COLLISION
1799 REM
1800 POSITION JJP,JJR:GET #1,X:POSITION
      JJP,JJR:PUT #1,X
1810 POSITION JJP+1,JJR:GET #1,Y:POSITION
      JJP+1,JJR:PUT #1,Y
```

```
1820 IF X<>32 OR Y<>32 THEN POP :GOTO 900
1830 POSITION JJP,JJR:PRINT JJ$;
1840 RETURN
1990 REM
2000 REM DATA FOR "CUSTOMIZED" CHARS.
2005 REM 8X8 MATRIX, TOP TO BOT.
2060 REM
2070 REM "JACQUE" HIMSELF
2080 REM
2090 DATA 24,60,24,126,90,90,36,36
2100 REM
2110 REM "ERIN"
2120 REM
2130 DATA 24,60,24,60,90,90,36,36
2140 REM
2150 REM MONSTER
2160 REM
2170 DATA 60,126,24,126,153,60,90,36
2330 REM
2490 REM ALWAYS LAST DATA STATEMENT
2495 REM
2500 DATA -1
6000 SOUND 0,4,8,15:FOR T=0 TO 3:NEXT T
6010 SOUND 0,8,8,15:FOR T=0 TO 3:NEXT T
6020 SOUND 0,0,0,0
6030 RETURN
8000 GRAPHICS 0:POKE 82,4
8005 POSITION 7,1
```



```
8010 ? "*** CAVERNS OF CARNAGE ***"  
8100 ?  
8110 ? "PRESS < TO MOVE LEFT OR"  
8120 ? "> TO MOVE JACQUE AND ERIN RIGHT"  
8130 ?  
8135 ? "PRESS SPACE-BAR TO FIRE POSITRON"  
8137 ? "BLASTER SHOTS TO HIT MONSTERS"  
8138 ?  
8140 ? "THE CLOSER THE MONSTER TO"  
8150 ? "JACQUE WHEN FINALLY HIT, THE"  
8160 ? "GREATER NUMBER OF POINTS ADDED"  
8930 ?  
8940 ? "PRESS ANY KEY TO BEGIN";  
8945 A=PEEK(764):IF A=255 THEN 8945  
8950 POKE 764,255:GOTO 1000
```

Variables Used in this Game Include:

BL\$=String of blanks
CHADR=Absolute memory address of alternate
char. set
CHBASE=Memory page of alternate char. set (* 256
= CHADR)
HIGHSC=High score
HIGHSC\$=High scorer's name
L=Width of left part of cavern wall
L\$=Cavern wall string
LJJP=Last position of couple
JJ\$=Jacque/Erin characters
JJP=Jacque/Erin column position

JJR=Jacque/Erin row position
M\$=Monster character
ML=Maximum "L" value
MPC=Missile position (column)
MPR=Missile row; = JJR if inactive
MS\$=Monster string to be printed
MSS\$=Initial monster string
SCORE=Obvious; the score
SPC\$=String of spaces
SW=Screen width
TC=Total # of game cycles
W=Cavern width
X=Miscellaneous variable

Explanation of the Program

2 Set graphics mode, turn off cursor
4 Clear keyboard, set margins
5-6 Set variables
7-8 Dimension strings
20-90 Set up alternate character set
95 Clear screen call
150-151 Set variables (JJ\$ = Erin/Jacque characters)
155-160 Set up inverse blank string
165 Initial meander position
170 Open "file" to access screen
180 Wipe cursor
185 Set initial couple position
190-195 Set up string of spaces
200-207 Same as 190-195

300 Increment cycle counter
502 Check for third cycle, perform action
510-555 Was third cycle, print new string with
 monster, at bottom of screen, automatic scroll
560 Check for Jacques/Erin collision
575 Check for missile collision
580-587 Update missile position if required
600-615 Print score, high score
690 Perform another cycle
800-802 Peek byte from keyboard, return if 255
 (no keys)
803 Re-initialize keyboard for next stroke
807-815 Adjust player's position if necessary
820-839 Fire missile if necessary and possible
900-980 Jacques/Erin hit something, end of
 game
1000-1020 Initialize screen
1010 Set up alternate character map base
1015 Set up appropriate colors
1100-1190 Check for missile hit, update score
1500-1520 Clear Jacques/Erin image and
 missile image
1710-1735 Check for missile hit
1770-1780 Missile hit, do update
1800-1840 Check for Jacques/Erin collision
 with object
2090 Data for "Jacques" character
2130 "Erin"
2170 "Monster"

2500 Data statement terminator

6000-6030 Sound effects when monster hit

8000-9000 Instructions

Some Changes You Might Like to Try:

1. Make other types of creatures appear and award different points.
2. Allow trapdoors in the cavern as an extra challenge to Jacque and Erin.
3. Allow safe side passages as an extra aid to Jacque and Erin.
4. If Jacque and Erin get over 10,000 points, you have managed to get the two through.

Escape

You are in the horrible prison of the Warlord and are attempting to escape with some of your fellow prisoners. But to escape, you must leave through the River of Death by passing through the Gates of Destruction.

In this game, you can help 100 prisoners to escape by guiding them through the gates. No one has ever got all 100 safely through. You guide each prisoner's boat down the river and through the gates, steering the boat left or right. Because the current is so swift, there's no way you can stop the boats. You must maneuver them just right to pass through the gates—if possible.

Watch out for the rocks at the edge of the river. They will also make you lose a prisoner if one touches them. You'll also notice that the shoreline changes as the game progresses.

This represents the tide exposing and covering some of the rocks at the edge. Good luck.

GAME PROGRAM LISTING

```
2 REM *** ESCAPE ***
5 RAD :GOSUB 8000
7 P=3
10 SW=38:M=6:TR=3:BR=18
12 K=6.28318/(6*P)
15 OPEN #1,4,0,"S:"
20 DIM L$(SW),SPC$(SW),G$(SW),BT$(1)
22 DIM IBT$(1)
25 TOTAL=100
30 FOR X=1 TO SW
35 L$(X,X)=CHR$(32+128):SPC$(X,X)=" "
37 G$(X,X)=CHR$(18)
40 NEXT X
42 TC=0:LEFT=TOTAL:BT$=CHR$(16)
43 ESCAPED=0:IBT$=CHR$(16+128)
45 C=INT(10*RND(0)):ROW=TR
50 BPC=SW/2+2:BPR=BR+5
60 GOSUB 4310
100 NB=INT(M*ABS(SIN(K*(ROW+C))))+1
103 SHWD=SW/2-NB+1
105 POSITION 1,ROW:POKE 752,1
110 PRINT L$(1,SW/2-NB);SPC$(1,NB*2);
    L$(1,SW/2-NB);:POKE 752,1
120 IF ROW=BPR AND BPC<=SW/2-NB
    THEN 3000
190 GW=1
200 POSITION SHWD+GW-1,ROW:GET #1,X
```

```
201 POSITION SHWD+GW-1,ROW:PUT #1,X
203 IF X=16 THEN POSITION BPC,BPR:
    ? " ";;GOTO 3000
205 POSITION SHWD,ROW:? G$(1,GW);
207 POSITION BPC,BPR:? BT$;
210 POKE 752,1:TC=TC+1:GW=GW+0.25
215 SOUND 0,250-SHWD-GW*8,10,2+GW
220 GOSUB 2000
230 IF TC/4=INT(TC/4) THEN POSITION
    BPC,BPR:PRINT " ";;BPR=BPR-1:
    GOSUB 2070
240 IF BPR=TR-1 THEN 4000
300 IF INT(GW)<NB*2 THEN 200
400 FOR X=15 TO 0 STEP -1
405 SOUND 0,255*RND(0),0,X
410 FOR T=0 TO 0:NEXT T
415 NEXT X
420 SOUND 0,0,0,0
1000 GOSUB 1500
1020 GOTO 100
1500 POSITION SHWD,ROW:? SPC$(1,NB*2);
1505 ROW=ROW+1
1510 IF ROW>BR THEN ROW=TR:C=C+1
1520 RETURN
2000 A=PEEK(764):IF A=255 THEN RETURN
2010 POKE 764,255
2015 LBPC=BPC
2020 BPC=PBC+(A=55)-(A=54)
```

```
2030 IF BPC<1 THEN BPC=1
2040 IF BPC>SW-2 THEN BPC=SW-2
2050 IF LBPC=BPC THEN RETURN
2060 POSITION LBPC,BPR:PRINT " ";
2070 POSITION BPC,BPR:GET #1,X
2080 POSITION BPC,BPR:PUT #1,X
2090 IF X<>32 THEN POP :GOTO 3000
2100 POSITION BPC,BPR:PRINT BT$;:RETURN
3000 POSITION 11,1:PRINT "ANOTHER
      ONE LOST!";
3003 POSITION BPC,BPR:GET #1,X
3005 POSITION BPC,BPR:? IBT$;
3010 SOUND 0,64,10,8:FOR T=0 TO 30
3015 NEXT T
3020 POSITION BPC,BPR:? BT$;
3030 SOUND 0,128,10,8:FOR T=0 TO 30
3035 NEXT T
3040 POSITION BPC,BPR:? IBT$;
3080 SOUND 0,0,0,0
3090 LEFT=LEFT-1
3100 FOR T=0 TO 300:NEXT T
3110 POSITION BPC,BPR:PUT #1,X
3120 GOTO 4110
4000 POSITION 14,1:PRINT "FREEDOM!!!!!";
4010 FOR L=1 TO 3
4015 SOUND 0,33,10,15:FOR T=0 TO 10
4020 NEXT T:SOUND 0,0,0,0
4025 FOR T=0 TO 10:NEXT T
```

```
4030 NEXT L
4040 SOUND 0,25,10,15:FOR T=0 TO 50
4050 NEXT T:SOUND 0,0,0,0
4090 LEFT=LEFT-1:ESCAPED=ESCAPED+1
4100 FOR T=0 TO 300:NEXT T
4110 FOR X=0 TO 2:POSITION 1,X
4120 ? SPC$;:NEXT X
4130 GOSUB 4310
4200 BPC=SW/2+2:BPR=BR+5:GOTO 300
4310 POSITION 1,0:? "ESCAPED: ";ESCAPED;
4320 POSITION 23,0:? "REMAINING: ";LEFT;
4325 IF LEFT=0 THEN 7000
4330 RETURN
5000 GRAPHICS 0:POKE 752,1
5010 POKE 82,1:POKE 83,39:RETURN
7000 FOR X=1 TO 5:POSITION 0,X
7005 PRINT SPC$;:NEXT X
7007 POSITION 12,2
7010 ? "*** GAME OVER ***"
7100 POSITION 9,4
7200 ? "CARE FOR ANOTHER GAME";
7300 INPUT BT$
7400 IF BT$<>"Y" THEN GRAPHICS 0:END
7500 GOSUB 5000:GOTO 42
8000 GOSUB 5000
8005 POSITION 8,0
8010 ? "***** E S C A P E *****"
8020 POKE 82,3
```

8100 POSITION 8,4
8110 ? "THE OBJECT OF THE GAME IS TO"
8120 ? "STEER THE BOAT () TO SAFETY."
8130 ? "THE CURRENT INEXORABLY
CARRIES YOU"
8140 ? "UP THE SCREEN TO YOUR GOAL OF"
8150 ? "FINAL ESCAPE AT THE TOP. PRESS"
8155 ? "< AND > KEYS TO MOVE LEFT
AND RIGHT."
8160 ? "AVOID CLOSING GATES, OR THE"
8170 ? "SHORELINE. CRASHING INTO EITHER"
8180 ? "ONE DESTROYS THE BOAT
AND DROWNS"
8190 ? "THE UNFORTUNATE OCCUPANT."
8195 ? "YOUR SCORE IS EQUAL TO THE"
8197 ? "NUMBER OF PEOPLE THAT YOU FREE."
8200 ?
8210 ? "PRESS ANY KEY WHEN READY";
8220 A=PEEK(764):IF A=255 THEN 8220
8230 POKE 764,255:GOTO 5000

Variables Used in this Game Include:

BPC=Boat column

BPR=Boat row

BR=Bottom row at which river plotted

BT\$=Boat character; input char. at end
of program

C=Offset in NB function for variable
river width

CHADR=Absolute memory address of alternate character set

CHBASE=Memory page of alternate char. set (* 256 = CHADR)

ESCAPED=Number of prisoners that have escaped

G\$=Gate characters

GW=Width of gate at any particular time

IBT\$=Permanently boat character

JJ\$=Jacque/Erin string

JJP=Jacque/Erin position

LEFT=# of prisoners left

LBPC=Last boat column

LBPR=Last boat row

L\$=River shore characters

NB=Number of blanks in a particular row of shoreline

P=Affects number of "waves" in shoreline shown at any one time on screen

RI\$=String of rocks

ROW=Current gate row

SHWD=Width of left shoreline (at any particular plot)

SPC\$=String of spaces

SW=Screen width

TC=Total # of game cycles

TOTAL=Total number of "prisoners"

TR=Top row at which shore plotted

X=Miscellaneous variable

Explanation of the Program

- 5 Set radian mode / initialize screen
- 7-12 Set up variables
- 15 Open "File" for screen access
- 20-22 Dimension strings
- 25 Set total number of prisoners
- 30-40 Set up strings
- 42-43 Set up variables
- 45 Random setting of river width / initial row
- 50 Initial boat position
- 60 Print "Escaped," etc., messages
- 100 Function for 1/2 river width (number of
blanks)
- 103 Width of left part of river
- 105-110 Print river section
- 120 Boat destroyed if collision with river bank
- 190 Initial gate width
- 200-203 Check for gate closure on boat
- 205 Print gate
- 207 Print boat
- 210 Erase cursor, increment cycle counter,
increment gate width counter
- 215 Ominous noise of gate closing
- 220 Check for key action
- 230 If every 4th cycle (one gate closure), move
boat up screen. Note that the 4 in "TC/4" is
the reciprocal of the 0.25 added to GW in line
210. To slow the speed at which the boat
moves up the screen, multiply 4 by some

integer (2, 3, etc.) To speed the boat up, change "4" to 2 or 1. Of course, changing 0.25 to some other value will also change the game (if greater than 0.25, such as 0.5, then gate will close faster, if smaller then gate will close more slowly, etc.) To change the speed but still have the boat going up the screen at one row per gate closure, the value added to GW in line 210 and the divisor of TC in line 230 must be reciprocals of each other.

- 240 If boat past top of river, freedom
- 300 If gate not yet closed, branch
- 400-420 "Crash" of gate closing
- 1000 Call to Clear gate, start at next or top row
- 1020 Loop to start next cycle
- 1500-1520 See line 1000
- 2000-2040 Update position of boat based on
keys
- 2050 If nothing done, return
- 2060-2100 Check for boat collision
- 3000-3120 Boat collided; make sound effects,
blink boat, decrement "LEFT," start another
prisoner
- 4000-4130 Boat freed; make sound effects, print
"FREEDOM" message, decrement "LEFT"
and increment "ESCAPED," start another
prisoner
- 4310-4330 Print # escaped, remaining
- 5000-5010 Initialize screen
- 7000-7500 Game over routine
- 8000-8300 Instructions

Some Changes You Might Like to Try:

1. Allow multiple gates to make it harder for the prisoners to get through.
2. Allow the prisoners' boats to temporarily stop.
3. Make rocks randomly appear in the water to make it harder for the boats to get through.

Rockfall!

Things are bad. The Wizard's creatures are attacking, and rocks are crashing down. How long can Jacque and Erin survive?

You can fire Jacque's blaster right or left at the creatures or up at the rocks. You score more points the closer you let the rocks or creatures come before blasting them.

If a rock hits Jacque or Erin, the game is over. Likewise, if a creature gets to either of them, the game ends.

GAME PROGRAM LISTING

```
1 REM *** ROCKFALL ***
2 GRAPHICS 1:POSITION 4,4:
  ? #6;"INITIALIZING";
3 REM CLEAR KEYBOARD,LEFT MARGIN
4 POKE 764,255:SW=20
5 HIGHSC=0:BASE=34
7 DIM JJ$(2),RI$(SW),R$(SW),SPC$(SW)
8 DIM SM$(1),HIGHNM$(10),LM$(1),RM$(1)
9 DIM RK$(1),DSP$(SW)
10 JJ$(1)=CHR$(BASE+98)
11 JJ$(2)=CHR$(BASE+1)
```



```
12 SM$(1)=CHR$(BASE+101)
13 LM$(1)=CHR$(BASE+3)
14 RM$(1)=CHR$(BASE)
15 RK$(1)=CHR$(BASE+4)
16 FOR X=1 TO SW:SPC$(X)=" "
18 RI$(X)="&":NEXT X
20 STDSET=57344:CHBASE=PEEK(15)+28
30 CHBASE=4*INT(CHBASE/4)+4
40 CHADR=CHBASE*256
50 FOR X=0 TO 767
60 POKE CHADR+X,PEEK(STDSET+X)
70 NEXT X
80 P=CHADR+8*(BASE-32)
90 READ X:IF X>-1 THEN POKE P,X:P=P+1:
  GOTO 90
95 GOSUB 8000
100 REM -----
105 REM
110 REM TC=TOTAL # GAME CYCLES
120 REM CRR=CURRENT ROCK ROW
125 REM LMP=LEFT MONSTER POSITION
126 REM RMP=RIGHT MONSTER POS.
130 REM JJP=COUPLE'S POSITION
131 REM MPR=UPWARDS MISSILE ROW
132 REM MPC=UPWARDS MISSILE COL.
133 REM SMP=SIDE. MISSILE POS.
134 REM SMV=SIDE. MISSILE VECTOR
135 REM LMPR=LAST UP MISSILE ROW
136 REM LMPC=LAST UP MISSILE COL.
137 REM LSMP=LAST SIDE MISSILE POS.
140 REM
150 TC=0
155 ICRR=5
165 SCORE=0
170 REM -----
```

```
175 GOSUB 1000
180 MPR=19:SMV=0:SMP=-1
185 LSMP=SMP:LMPR=MPR:LMPC=MPC
190 POKE 764,255
199 REM
200 REM BEGIN (ANOTHER) MAIN CYCLE
210 REM (ROCKS SUCCESSFULLY SHOT,
220 REM NOBODY CRUSHED)
230 REM
240 TC=TC+1:LCNT=0
245 DMP=1+(TC>2)+2*(TC>6)+4*(TC>10)
250 R$=RIS$
255 FOR X=1 TO 7-DMP:Y=1+18*RND(0)
260 R$(Y,Y)=" ":NEXT X
330 LMP=1+INT(DMP*RND(0))
340 RMP=19-INT(DMP*RND(0))
350 JJP=10
355 CRR=ICRR+DMP
365 REM
366 REM -----
367 REM
370 LCNT=LCNT+1
371 REM
372 POSITION 0,0:PRINT #6;"score: ";SCORE
373 POSITION 0,1:PRINT #6;" high: ";HIGHSC::
    IF HIGHNM$<>"" THEN PRINT #6;
    "(BY ";HIGHNM$;")";
374 REM
375 QUANT=ABS(LCNT/(8-DMP)-INT
    (LCNT/(8-DMP)))
376 GOSUB 1100
377 LSMP=SMP:SMP=SMP+SMV
378 SMP=SMP*(SMP<20)-(SMP>19)
379 SMV=SMV*(SMP>-1)*(SMP<20)
380 REM
```

```
381 IF SMP>-1 THEN POSITION LSMP,19:? #6;  
    " ";;POSITION SMP,19:? #6;SM$;:GOSUB 1100  
390 IF LMP>-1 THEN LMP=LMP+(DMP>5*RND(0))  
391 IF RMP>-1 THEN RMP=RMP-(DMP>5*RND(0))  
395 IF LMP<0 THEN IF DMP>5*RND(0) THEN  
    LMP=DMP  
397 IF RMP<0 THEN IF DMP>5*RND(0) THEN  
    RMP=19-DMP  
403 REM  
404 REM UPDATE, DRAW ROCKS  
405 REM  
406 POSITION 0,CRR:IF QUANT<1E-06 THEN  
    SOUND 0,50,10,8:PRINT #6;SPC$;  
    CRR=CRR+1:GOSUB 1100:SOUND 0,0,0,0  
407 POSITION 0,CRR:PRINT #6;R$;  
408 REM BRANCH IF ROCKS AT BOTTOM  
409 IF CRR=19 THEN 750  
410 REM  
415 DSP$=SPC$  
416 REM  
420 DSP$(JJP+1,JJP+2)=JJ$  
421 REM  
422 IF LMPCR<19 THEN POSITION LMPCR,  
    LMPCR:PUT #6,32  
425 IF SMP>-1 THEN DSP$(SMP+1,SMP+1)=SM$  
430 IF RMP>-1 THEN DSP$(RMP+1,RMP+1)=RM$  
432 IF LMP>-1 THEN DSP$(LMP+1,LMP+1)=LM$  
435 IF MPCR<19 THEN POSITION MPC,MPCR:  
    PUT #6,138  
437 POSITION 0,19:? #6;DSP$;  
439 REM  
440 REM UPWARDS MISSILE UPDATE  
441 REM  
442 LMPCR=MPCR:LMPCR=MPCR:REM SAVE POS.  
445 MPCR=19*((MPCR=19) OR (MPCR=0))+  
    (MPCR<19)*(MPCR>0)*(MPCR-1)
```

```
459 REM
460 REM CHECK FOR MISSILE HIT(S)
461 REM
465 GOSUB 1100
499 REM
500 REM CHECK FOR MONSTER/JJ HIT
501 REM
505 IF JJP<>LMP AND (JJP+1)<>RMP THEN 610
506 REM COLLISION WITH MONSTER
507 REM
510 POSITION 5,4:PRINT #6;"** MUNCH **";
    GOTO 760
511 REM
599 REM
600 REM PERFORM ACTION FROM KEYS
605 REM
610 GOSUB 800
690 GOTO 370
699 REM _____
749 REM ROCKS AT BOTTOM, DO CHECKS
750 IF R$(JJP+1,JJP+2)=" " THEN 775
755 POSITION 5,4:PRINT #6;"** CRUNCH **";
760 POSITION 6,6:PRINT #6;"GAME OVER";
762 IF SCORE<HIGHSC THEN 765
763 PRINT CHR$(125);"NEW HIGH SCORE!! YOUR
    NAME";:INPUT HIGHNM$
764 HIGHSC=SCORE
765 PRINT CHR$(125);"CARE FOR ANOTHER
    GAME?";:INPUT R$:IF R$(1,1)="Y" THEN 150
770 GRAPHICS 0:END
774 REM PLAYER MADE IT
775 POSITION JJP,19:? #6;JJ$;
777 POSITION 4,5:? #6;"YOU MADE IT";
778 FOR T=1 TO 300:NEXT T
780 GOTO 175
```

```
795 REM -----
800 A=PEEK(764):IF A=255 THEN RETURN
803 POKE 764,255
805 REM ADJUST PLAYER'S POSITION
810 JJP=JJP-(A=54)+(A=55)
812 IF JJP<3 THEN JJP=3
813 IF JJP>16 THEN JJP=16
815 REM FIRE UP MISSILE?
820 IF A=33 THEN IF MPR=19 THEN
    MPR=18:MPC=JJP+1
825 REM FIRE SIDE MISSILE?
827 IF (A=10)*(SMP=-1) THEN SMP=JJP-1:
    SMV=-1:POSITION SMP,19:? #6;SM$;
830 IF (A=15)*(SMP=-1) THEN SMP=JJP+2:
    SMV=1:POSITION SMP,19:? #6;SM$;
839 RETURN
840 REM -----
999 END
1000 GRAPHICS 1:SETCOLOR 1,4,14
1010 POKE 756,CHBASE:POKE 82,0
1020 RETURN
1097 REM
1098 REM SIDE MISSILE HIT MONSTER?
1099 REM
1100 IF SMP=LMP THEN LMP=-2:GOSUB 1200
1110 IF SMP=RMP THEN RMP=-2:GOSUB 1200
1115 IF MPR<>CRR THEN 1130
1120 IF R$(MPC+1,MPC+1)<>" " THEN
    R$(MPC+1,MPC+1)=" ":GOSUB 6000:MPR=19
1130 RETURN
1199 REM
1200 SOUND 0,192,10,8:SOUND 1,255,0,8
1205 POSITION SMP,19:? #6;"*";
1210 SMV=0:SMP=-1:SCORE=SCORE+10
1215 FOR T=1 TO 2:NEXT T
```



```
1220 SOUND 0,0,0,0:SOUND 1,0,0,0
1230 RETURN
1990 REM
2000 REM DATA FOR "CUSTOMIZED" CHARS.
2005 REM 8X8 MATRIX, TOP TO BOT.
2010 REM
2020 REM LEFT MOVING MONSTER
2040 REM
2050 DATA 8,20,10,50,2,14,17,33
2060 REM
2070 REM "JACQUE" HIMSELF
2080 REM
2090 DATA 24,60,24,126,90,90,36,36
2100 REM
2110 REM "ERIN"
2120 REM
2130 DATA 24,60,24,60,90,90,36,36
2140 REM
2150 REM
2160 REM RIGHT MOVING MONSTER
2170 REM
2180 DATA 16,40,80,76,64,112,136,132
2230 REM
2240 REM ROCK
2250 REM
2260 DATA 24,60,126,127,127,126,62,60
2290 REM
2300 REM BULLET
2310 REM
2320 DATA 0,0,0,24,0,0,0,0
2330 REM
2490 REM ALWAYS LAST DATA STATEMENT
2500 DATA -1
5000 POKE 764,255
5010 A=PEEK(764):IF A=255 THEN 5010
```

```
5015 POKE 764,255
5020 PRINT A
5040 GOTO 5010
6000 SOUND 0,16,8,15:FOR T=0 TO 3:NEXT T
6010 SOUND 0,32,8,15:FOR T=0 TO 3:NEXT T
6020 SOUND 0,0,0,0
6030 RETURN
8000 GRAPHICS 0:POKE 82,2
8010 POSITION 11,1
8020 ? "*** ROCKFALL ***"
8030 ?
8040 ? "THE GOAL OF THE GAME IS TO AVOID"
8050 ? "DESTRUCTION BY EITHER BEING
      CRUSHED"
8060 ? "BY FALLING ROCKS OR MUNCHEDED
      ON BY"
8070 ? "INCONSIDERATE MONSTERS. THESE
      FATAL"
8080 ? "THINGS CAN BE AVOIDED BY
      APPROPRIATE"
8090 ? "USE OF POSITRON BLASTERS. A HEAVY"
8100 ? "ARTILLERY MODEL CAN BE SHOT
      UPWARDS"
8110 ? "BY PRESSING THE SPACE BAR, TO
      DESTROY"
8140 ? "FALLING ROCKS. AND, BECAUSE THE"
8150 ? "MONSTERS REFUSE TO LISTEN TO
      REASON,"
8160 ? "THEY ARE FORCIBLY DISSUADED WITH
      YOUR"
8170 ? "HAND MODEL POSITRON BLASTER —
      PRESS"
8180 ? "P TO SHOOT LEFT, = FOR RIGHT."
8190 ? "TO MOVE LEFT OR RIGHT, PRESS"
8200 ? "< OR >."
```

```
8210 ?  
8220 ? "PRESS ANY KEY TO BEGIN";  
8230 IF PEEK(764)=255 THEN 8230  
8240 POKE 82,0  
8300 POKE 764,255:GOTO 1000
```

Variables Used in this Game Include:

BASE=Alternate character code base
CRR=Current rock row
DMP=Difficulty function; the larger, the harder
 (currently set up so that it generally
 increases as TC increases)
DSP\$=Display string (at row 19)
HIGHSC=High score
HIGHSC\$=High scorer's name
ICRR=Base rock row
LCNT=Total # local (within main) cycles
LM\$=Left monster character
LMP=Left monster position (-2 if no
 left monster)
LMPR=Last up missile row
LMPC=Last up missile column
LSMP=Last side missile position
 JJP=Jacque/Erin position
 JJ\$=Characters for Erin/Jacque
MPC=Upwards missile column
MPR=Upwards missile row
QUANT=Very close to zero every (8-DMP)th
 count of LCNT (for determining action)
RI\$=String of rocks
RK\$=Individual rock character
RM\$=Right monster character

RMP=Right monster position (-2 if no
right monster)
SCORE=Obviously the score
SM\$=Side missile character
SMP=Side missile position
SMV=Side missile vector (-1 = left, 1 =
rightwards)
SPC\$=String of spaces
SW=Screen width
TC=Total # of game main cycles
X=Miscellaneous variable

Explanation of the Program

2 Initialize screen mode
4-5 Clear keyboard, set variables
7-9 Dimension strings
10-15 Set up individual strings
16-18 Set up blank, rock strings
20-90 Set up alternate character set
95 Display instructions
150-165 Set up variables
175 Initialize screen
180-185 Set up variables
190 Clear keyboard
240 Increment main counter, clear local counter
variables
245 Difficulty function (greater = harder)
250-260 Set up rock string
330-340 Initial monster positions
350 Initial couple position

355 Initial rock row
370 Increment local counter
372-373 Print score, high score
375 Action variable (very close to zero = (8-DMP)th count of local counter (LCNT))
376 Check for side missile hit
377-379 Update side missile position
381 Clear missile, redraw at new location
390-391 Update monster positions
395-397 Generate new monsters if necessary
406 Draw rocks at new position
407 Display rocks
409 Branch if rocks at bottom
415-437 Set up display string, display at bottom
442-445 Update upwards missile
465 Check for missile hit(s)
505 Monster get Jacque/Erin?
510 Yep.
610 Perform key action
690 Do another cycle
750-780 Rocks at bottom, either they got it (in which case end the game), or they didn't (in which case display message and start another main cycle)
800-803 Get keyboard char, return if nothing
810-813 Adjust Jacque/Erin position based on keys
820 Fire up missile if required and possible
827-839 Fire side missile if necessary and possible

1000-1020 Initialize screen, alternate character set
1100-1110 Check for side missile hit
1115-1120 Check for up missile hit
1200 Monster hit noise
1205 Monster hit explosion
1210-12230 Stop missile, update score, return
2050 Data for Left-moving monster
2090 Jacques image matrix
2130 Erin character
2180 Right-moving monster
2260 Rock
2320 Side missile (or bullet)
2500 Data terminator
6000-6030 Noise for rock destruction
8000-8300 Instructions

Some Changes You Might Like to Try:

1. Make the creatures appear and disappear so that it is harder to hit them.
2. Allow some colored rocks or Warbirds to appear and give more points for shooting them.

Timelost is the first book in a new unique series of integrated comic adventures and computer action games.

A teenage computer whiz and his young sister are swept through Time Holes and meet attacking Warbirds of the past, a mysterious man from the future, and a sinister Time Wizard.

Each lost-in-time adventure is combined with an arcade-type computer action game. Explanations help new users to learn programming and suggested variations challenge young users to create more game fun.



0-88022-056-2

